
ABSTRACT  
Intensive longitudinal data are increasingly obtained in health studies to examine subjective 
experiences within changing environmental contexts. Such studies often utilize ecological 
momentary assessment (EMA) and/or experience sampling methods to obtain up to 30 or 
40 observations for each subject within a period of a week or so. In this paper, we focus on 
data from an adolescent smoking study using ecological momentary assessment in which 
there was interest in examining mood variation associated with smoking. We describe the 
mixed-effects location scale (MELS) model which allows covariates to influence both the 
within-subject (WS) and between-subject (BS) variances, in addition to their influence on 
the mean. The model also includes a subject-level random effect to the within-subject 
variance specification. This permits subjects to have influence on the mean, or location, and 
variability, or (square of the) scale, of how soon the first smoking episode happened after 
wake-up. Additionally, we allow the location and scale random effects to be correlated. 
These mixed-effects location scale models have useful applications in many research areas 
where interest centers on the joint modeling of the mean and variance structure. In this 

presentation, we describe how SAS® PROC NLMIXED and SAS PROC MCMC can be used to 
estimate the parameters of the MELS model. 

INTRODUCTION  
Linear mixed effect models are frequently used for handling heterogeneity across clusters in 
multilevel longitudinal data. For a two-level mixed model for repeated measures (level-1) 
nested within-subjects (level-2), the random subject effect summarizes the latent subject-
specific features apart from the observed covariate effects of interest. In traditional data 
analysis on longitudinal data using mixed effect models, the subject-level random effect is 
often specified in the mean function of the repeated outcome measurements. By including 
the random subject effects in the mean function, subjects are allowed to have different 
mean levels for their outcome trajectories.  

The development of novel real-time data collection approaches such as ecological 
momentary assessments (EMA) even provides researchers intensive amount of longitudinal 
data which contains sufficient information to infer not only the subject mean but also the 
within-subject variability of the repeated outcome measurements. Motivated by these 
multilevel intensive longitudinal data, a mixed-effect location-scale (MELS) model was 
proposed as an extension of the linear mixed model (Hedeker, D. et al., 2008, 2013). In the 
MELS model, the random subject effect in the mean function of the outcome is called the 
random location effect; while the random subject effect in the within-subject variance 
function is called the random scale effect. Both are assumed to follow a bivariate normal 
distribution with mean zero and with a general variance-covariance matrix. The intuition of 
including the random scale effect is that, as is observed in many studies, some participants’ 

outcome, such as mood, may display a more consistent pattern but others’ mood can vary 
erratically across time. By including the scale effect, subjects are allowed to display different 
levels of variability in terms of their outcome trajectories. 

Although the outcome variable in MELS is assumed to follow a normal distribution, MELS 
belongs to the class of generalized linear mixed effect model (GLMM) as the relationship 
between the outcome and the random scale effect is not linear. Flexible estimation 
approaches should be used to handle this complex distribution assumption and specification. 
In SAS/STAT, the NLMIXED modelling procedure enables us to specify a flexible formulation 
for within-subject variability as well as a user-defined likelihood function. In NLMIXED, 
maximum likelihood estimation is used for estimation. A critical step of this estimation 
method is to integrate out the random location/scale effects from the full likelihood function 
to obtain a marginal likelihood containing only the fixed effects. In other words, this 
integrated marginal likelihood function is the objective function for later optimization. For 
GLMM, usually there is no closed-form solution for this integral evaluation so that more 
general integration techniques such as adaptive Gaussian quadrature can be used. 
However, a shortcoming of this algorithm is that the complexity grows exponentially with 
the dimension of the random effects and so it is often time-consuming and computationally 
demanding.  

With similar syntax as PROC NLMIXED, PROC MCMC also has flexibility for fitting 
unconventional generalized mixed effect models. Instead of maximizing the likelihood 
function, PROC MCMC utilizes a Bayesian approach to draw samples from the posterior 
distributions of the model parameters using the Metropolis-Hasting Algorithm. This 
procedure also produces the credible intervals or highest posterior density intervals for 
inference. According to our practice, the computation time of Markov Chain Monte Carlo 
methods doesn’t grow significantly along with the dimension of random effects.  

The paper is organized as follows. First, the MELS section introduces the notation of the 
MELS model, including transformation of the random effects. The next section 
“ADOLESCENT SMOKING SUDY” describes the real dataset that is used in the examples. 
Sections “ESTIMATE MELS USING PROC NLMIXED” and “ESTIMATE MELS USING PROC 
MCMC” provides syntax examples of PROC NLMIXED and PROC MCMC for estimating the 
parameters of the MELS model using the adolescent smoking data.  

MIXED-EFFECT LOCATION-SCALE (MELS) MODEL  
The model structure of the MELS model is similar to the linear mixed effect model. Assume 
the data are two-level and the outcome variable 𝑦  of subject 𝑖 (𝑖 = 1, 2, . . . , 𝑁 𝑠𝑢𝑏𝑗𝑒𝑐𝑡𝑠) at 
the 𝑗   assessment occasion within each subject  (𝑗 = 1, 2, . . . , 𝑛  𝑜𝑐𝑐𝑎𝑡𝑖𝑜𝑛𝑠) is expressed as 
Equation (1). 𝜷 is the (𝑝 + 1) -dimension vector of fixed effects and  𝜈  is the 𝑖   subject’s 
random location effect in the mean function of the outcome, which represents the additional 
deviation of the covariate effects/intercept of subject 𝑖 from the population 
 

𝑦 = 𝑿𝒊𝒋
𝑻 𝜷 + 𝜈 + 𝜖  (1)  

The variance of 𝜈 , σ  represents the between-subject (BS) variance. Instead of assuming 
it to be constant across subjects, σ  can be further expressed by an exponential form in the 

log-linear equation (2) to allow covariates 𝒖𝒊𝒋 to affect the BS variance. 𝜶 is the fixed effect 
vector for the BS variance associated with the 𝒖𝒊𝒋 covariates. The use of the exponential 
function ensures a positive value for the resulting variance. 
 
  

σ = 𝑒𝑥𝑝(𝒖  
𝑻 𝜶) (2)  

𝜖  in Equation (1) is the within-subject (WS) measurement error and is assumed to follow a 

normal distribution 𝑵 0,  σ .  Similar to Equation (2), the WS variance can be represented 

by a log-linear function (Equation (3)) as follows:  
 

σ = 𝑒𝑥𝑝(𝒘  
𝑻 𝝉 + 𝜔 ) (3)  

Here, 𝝉 is the fixed effect vector affecting WS variance corresponding to covariates 𝒘𝒊𝒋. 𝜔  is 
the 𝑖   subject’s random scale effect in the WS variance. By including 𝜔 , even subjects with 
the same covariate values can have different levels of variation in their longitudinal outcome 
trajectories.  

𝜈  and 𝜔  are not necessarily independent. Therefore, the distribution of (𝜈𝑖, 𝜔 ) is  

𝜈
𝜔 ~ 𝑁

0

0
,

σ 𝜎

𝜎  𝜎
 

where 𝜎  and 𝜎  are standard deviations of (𝜈𝑖, 𝜔 ), and 𝜎  is the covariance between the 

random location and scale effects.   

ADOLESCENT SMOKING STUDY 
We will use real data collected from participants in an Adolescent Smoking Study, which 
included 461 participants that were either 9th or 10th graders. In this study, participants 
carried a PDA for a week at baseline, and responded to random prompts, occurring 
approximately 4 to 5 times per day, and self-initiated smoking events.  These subjects were 
similarly measured at the 6-month, 15-month, 2nd year, 5th and 6th year follow-up waves. 
For this part of the study, we focused on a subset of 242 subjects (𝑁 = 242) that had data at 
years 5 and/or 6 and contributed at least two waves with ≥ 2 smoking events at each wave. 
Therefore, subjects can have different numbers of smoking days and the number of 
smoking days 𝑛   ranged from 2 to 18 with mean 8.37 for each subject (median=8). The 
longitudinal outcome of interest in this example is the time of the first report of smoking in 
the day (after 5am), which is a biomarker for smoking dependence. It is thought that 
subjects who smoke their first cigarette earlier in the day are more dependent smokers.  
From descriptive analysis, the outcome variable ranged from 0.02 to 23.13 hours (past 
5am) with a mean=9.55 hours. 
 
Conventionally, time-to-event data are analyzed by survival models. As we observed that 
the outcome was distributed close to a normal distribution, we assumed it to follow a 
normal distribution as specified in the MELS model of the last section. The outcome variable, 
time to the first cigarette of the day, is expressed in units of hours after 5am in the 
morning: 
 
TimeHour = timeT - 5; 
 
Before fitting models, we first preprocess some independent variables in the dataset. From 
the weekday variable (1 to 7), seven binary dummies were created.  Six of these seven 
dummies will be used in the analysis.   
 
mon=0;tue=0;wed=0;thu=0;fri=0;sat=0;sun=0; 
if wkday5am=1 then mon=1; 
if wkday5am=2 then tue=1; 
if wkday5am=3 then wed=1; 
if wkday5am=4 then thu=1; 
if wkday5am=5 then fri=1; 

if wkday5am=6 then sat=1; 
if wkday5am=7 then sun=1; 

 

MELS MODEL ESTIMATION USING PROC NLMIXED 
As PROC NLMIXED can be sensitive to the specification of initial values, we first fit a 
heterogeneity mixed model using PROC MIXED and use the parameter estimates as starting 
values for many of the parameters of our MELS model (that will be fit with PROC NLMIXED).  

 
PROC MIXED METHOD=ML COVTEST; 
  CLASS id;  
  MODEL TimeHour=tue wed thu fri sat sun wave6 Smk7rate/s; 
  RANDOM INT/SUBJECT=id;     
  REPEATED/LOCAL=exp(tue wed thu fri sat sun wave6 Smk7rate); 
RUN; 
 
The PROC MIXED procedure is fitting the model as below.  

 The mean function 𝑦 = (𝛽 + Σ  𝛽 𝐷 + 𝛽 𝑤𝑎𝑣𝑒 + 𝛽 𝑆𝑚𝑘𝑅𝑎𝑡𝑒 ) + 𝜈 + 𝜖   

 For WS variance, σ = 𝐸𝑋𝑃(𝜏0 + Σ𝑘=1 
6 𝜏𝑘𝐷𝑖𝑗𝑘 + 𝜏7𝑤𝑎𝑣𝑒𝑗 + 𝜏8𝑆𝑚𝑘𝑅𝑎𝑡𝑒𝑖𝑗); note that PROC 

MIXED does not allow a random scale effect.  

 For BS variance, σ = 𝐸𝑋𝑃(𝛼 ) and contain only the intercept. 

The output for the variance parameters from PROC MIXED is displayed below: 

 

Display 1. PROC MIXED: BS And WS Variance 

Estimates with initial of ‘EXP’ denotes the estimates for 𝝉  to 𝝉 . The term “Residual” 
denotes the estimate of 𝐸𝑋𝑃(𝜏 ), and the term “Intercept” denotes the variance of the 
location random effect, the BS variance, namely 𝜎 = 𝐸𝑋𝑃(𝛼 ). Therefore, the estimate for 
𝜏  is ln(17.6821) = 2.8726 and the estimate for  𝛼  is ln(4.1784) = 1.43 which can be used as the 
initial values for the parameters in the BS variance as well as the WS variance in the 
subsequent PROC NLMIXED code. 

Display 2. PROC MIXED: Fixed Effects 

Similarly, the estimated displayed in the output above provides starting values for the 
regression coefficients 𝜷. 

These PROC MIXED estimates are then used as starting values for 𝜷 as well as 𝝉, in other 
words the fixed effects in the mean function and the WS variance function. However, we 
only obtain the intercept in the BS variance function from PROC MIXED. Therefore, we plug 
in the estimates of 𝜷 𝝉 and 𝛼  from PROC MIXED as starting values in the PROC NLMIXED 
PARMS statement, and assign values of zero for the few remaining parameters.  

 
PARMS beta0=11.4756 beta_tue=-0.6326 beta_wed=-0.7576 beta_thu=-0.6899 
beta_fri=-0.5676 beta_sat=-0.04530 beta_sun=0.4450 beta_wave6=-0.2306 
beta_smk7rate=-1.3321  
   
tau0=2.8726 tau_tue=-0.1332 tau_wed=-0.2225 tau_thu=-0.02743  
tau_fri=0.003630 tau_sat=0.04770 tau_sun=0.01055 tau_wave6=0.03296 
tau_smk7rate=-0.1792  
 
alpha0=1.43 alpha_tue=0 alpha_wed=0 alpha_thu=0 alpha_fri=0 alpha_sat=0 
alpha_sun=0 alpha_wave6=0 alpha_Smk7rate=0 
 
corr=0 ln_varScale=0; 

 

For estimation, PROC NLMIXED uses maximum likelihood estimation (MLE) method.  

PROC NLMIXED GCONV=1e-12 TECH=QUANEW METHOD=GAUSS QPOINTS=11; 
 

Prior to the optimization stage, full conditional likelihood will be integrated with respect to 
random effects and then the resulted marginal likelihood will be the objective function for 
optimization. The Adaptive Gaussian Quadrature method (METHOD=GAUSS) is a general 
approach for this integration step and is applicable for generalized linear models and 11 
quadrature points are specified for this integral evaluation. For NLMIXED, the default 
optimization method is Quasi-Newton method (TECH=QUANEW) which uses the gradient 
function to approximate the Hessian matrix so that no extra steps are needed for computing 
the Hessian. As a result, the computation time for each iteration of the Quasi-Newton 
method is faster than the traditional Newton-Raphson algorithm, however more steps may 
be needed for convergence.  

For the efficiency of estimation, the random effects (𝜈𝑖, 𝜔 ) can be represented in terms of 
standardized random effects  (𝜃 , 𝜃 ) with unit variance and 𝜌  as the correlation. In other 
words, (𝜈𝑖, 𝜔 ) =  (𝜎 𝜃 , 𝜎 𝜃 ) and (𝜃 , 𝜃 ) follow a bivariate normal distribution as follows: 

𝜃
𝜃

~ 𝑁
0

0
,

1 𝜌
𝜌  1

(4) 

Using this standardization, the integral evaluation step is simplified so that the computation 
time taken for the quadrature method will be shorter and convergence will be easier to 
achieve. We could also use the Cholesky decomposition (which is what we will use in the 
PROC MCMC section below).  

The outcome 𝑦  is assumed to follow a normal distribution 𝑁(𝜇 , σ ). 

MODEL TimeHour ~ NORMAL(mu,varWS);  
 

For the BS variance, σ = 𝐸𝑋𝑃(𝛼 + Σ  𝛼 𝐷 + 𝛼 𝑤𝑎𝑣𝑒 + 𝛼 𝑆𝑚𝑘𝑅𝑎𝑡𝑒 ),  

varBS = EXP(alpha0 + alpha_tue*tue + alpha_wed*wed + alpha_thu*thu + 
alpha_fri*fri + alpha_sat*sat + alpha_sun*sun + alpha_wave6*wave6 + 
alpha_Smk7rate*Smk7rate);  

 

For the mean function 𝜇 = 𝛽0 + Σ𝑘=1 
6 𝛽𝑘𝐷𝑖𝑗𝑘 + 𝛽7𝑤𝑎𝑣𝑒𝑗 + 𝛽8𝑆𝑚𝑘𝑅𝑎𝑡𝑒𝑖𝑗 + 𝜈𝑖 and 𝜈 = 𝜎 𝜃  

nu = SQRT(varBS)*theta1; 
 
mu = beta0 + beta_tue*tue + beta_wed*wed + beta_thu*thu + beta_fri*fri + 
beta_sat*sat + beta_sun*sun + beta_wave6*wave6 + beta_Smk7rate*Smk7rate + 
nu; 

 

For the WS variance, σ = 𝐸𝑋𝑃(𝜏0 + Σ𝑘=1 
6 𝜏𝑘𝐷𝑖𝑗𝑘 + 𝜏7𝑤𝑎𝑣𝑒𝑗 + 𝜏8𝑆𝑚𝑘𝑅𝑎𝑡𝑒𝑖𝑗 + 𝜔 ) and 𝜔 = 𝜎 𝜃2𝑖  

StdDev_scale=SQRT(EXP(ln_varScale));  
omega= StdDev_scale*theta2;  

 
varWS = EXP(tau0 + tau_tue*tue + tau_wed*wed + tau_thu*thu + tau_fri*fri + 
tau_sat*sat + tau_sun*sun + tau_wave6*wave6 + tau_Smk7rate*Smk7rate + 
omega);  

 
In the above, we represent 𝜎  as an exponential function (which is always positive) so that 
this optimization is free of any constraints. The standard deviation of the scale random 
effect 𝜔  is the square-root of this exponential function.  

Finally, the random effect distribution is assumed to follow s bivariate normal distribution 
(Equation (4)). 

RANDOM theta1 theta2 ~ NORMAL([0,0], [1,corr,1]) SUBJECT=id;  
 
Combining these statements together provides the PROC NLMIXED code for estimation of 
the parameters of the MELS model: 

PROC NLMIXED GCONV=1e-12 TECH=QUANEW METHOD=GAUSS QPOINTS=11; 
 

PARMS beta0=11.4756 beta_tue=-0.6326 beta_wed=-0.7576 beta_thu=-0.6899 
beta_fri=-0.5676 beta_sat=-0.04530 beta_sun=0.4450 beta_wave6=-0.2306 
beta_smk7rate=-1.3321  

tau0=2.8726 tau_tue=-0.1332 tau_wed=-0.2225 tau_thu=-0.02743  
tau_fri=0.003630 tau_sat=0.04770 tau_sun=0.01055 tau_wave6=0.03296 
tau_smk7rate=-0.1792  
 
alpha0=1.43 alpha_tue=0 alpha_wed=0 alpha_thu=0 alpha_fri=0 alpha_sat=0 
alpha_sun=0 alpha_wave6=0 alpha_Smk7rate=0 
 
corr=0 ln_varScale=0; 
 
varBS = EXP(alpha0 + alpha_tue*tue + alpha_wed*wed + alpha_thu*thu + 
alpha_fri*fri + alpha_sat*sat + alpha_sun*sun + alpha_wave6*wave6 + 
alpha_Smk7rate*Smk7rate); 
 
StdDev_scale=sqrt(exp(ln_varScale)); 
 
nu= SQRT(varBS)*theta1; 
omega= StdDev_scale*theta2; 
 
mu  = beta0 + beta_tue*tue + beta_wed*wed + beta_thu*thu + beta_fri*fri  + 
beta_sat*sat + beta_sun*sun + beta_wave6*wave6 + beta_Smk7rate*Smk7rate + 
nu; 
 
varWS = EXP(tau0 + tau_tue*tue + tau_wed*wed + tau_thu*thu + tau_fri*fri   
+ tau_sat*sat + tau_sun*sun + tau_wave6*wave6 + tau_Smk7rate*Smk7rate + 
omega); 

 
MODEL TimeHour ~ NORMAL(mu,varWS); 
RANDOM theta1 theta2 ~ NORMAL([0,0], [1,corr,1]) SUBJECT=id; 
RUN; 
 
The parameter estimates from PROC NLMIXED are listed below. We conclude that, 
controlling for other covariate effects, higher smoking rate is significantly associated with 
earlier first cigarette (mean estimate = 1.46 hours).  In terms of the WS variance, higher 
smoking rate is associated with greater consistency (i.e., lower WS variance).  
Exponentiating the estimate (exp(-0.2) = 0.82) yields a variance ratio estimate; with every 
unit increase in smoking rate the WS variance is reduced by 18%.  Thus, subjects with 
higher smoking rates have their first cigarette earlier, on average, and exhibit greater 
consistency in the time for their first cigarette.   Smoking rate does not have a significant 
effect on the BS variance.  Another interesting observation is that the correlation between 
location and scale is positive, which indicates that subjects who smoke the first cigarette 
earlier/later in the morning tend are more consistent/erratic.  

Display 3. PROC NLMIXED: Solutions 

The computation time for this PROC NLMIXED analysis took almost 26 minutes for 
convergence. One reason is because a relatively large number (11) of quadrature points 
were specified. For faster computation, we can specify only 1 quadrature point in most of 
the cases which is equivalent to the Laplace Approximation (Pinheiro, J. C., Chao, E. C., 
2016). Alternatively, when we did not specify the QPOINTS, then PROC NLMIXED 
determined that only 5 quadrature points were needed for this example and the 
computation time was about 6 minutes. 

MELS MODEL ESTIMATION USING PROC MCMC 
In PROC MCMC, which uses Bayesian methods, the parameters to be estimated are viewed 
as random variables and are assumed to follow certain probabilistic distributions. PROC 
MCMC uses Markov chain Monte Carlo sampling to generate sequential samples from their 
posterior distribution given the data. The MCMC method does not include integration steps 
so that the complexity of this method will not increase dramatically with the dimension of 
the random effects.    

The syntax of PROC MCMC is similar to PROC NLMIXED, with some additional specifications. 
Here, we specify 100,000 repetitions of MCMC sampling (NMC). The first few repetitions are 
called the burn-in samples which can be highly inter-dependent before the Markov Chain 
converges. In this example, these 500 burn-in samples (NBI=500) will be excluded from the 
calculation of posterior sample means. Thinning is used to select the posterior samples for 
the mean calculation. Here, we specify THIN=25 so that one sample is selected from every 
25 repetitions for the mean calculation. In the Bayesian approach, the frequentist model 
selection criterion of AIC or BIC are not provided, instead the Deviance Information 
Criterion (DIC) will be provided. 

 
PROC MCMC SEED=9879 NMC=100000 NBI=500 MAXTUNE=50 THIN=25 DIC; 
 
Similar to PROC NLMIXED, we also specify the initial values using the estimates from PROC 
MIXED. It is convenient that PROC MCMC can do both the group-wise update and 
parameter-wise update. Multiple PARMS statement are allowed and each PARMS statement 
represents a block of parameters to be updated at the same time. Here we only use one 
PARMS statement which means one single block and all of the model parameters are 
updated at each step. 

 
PARMS beta0=11.4756 beta_tue=-0.6326 beta_wed=-0.7576 beta_thu=-0.6899 
beta_fri=-0.5676 beta_sat=-0.04530 beta_sun=0.4450 beta_wave6=-0.2306 
beta_smk7rate=-1.3321  
 
tau0=2.8726 tau_tue=-0.1332 tau_wed=-0.2225 tau_thu=-0.02743  
tau_fri=0.003630 tau_sat=0.04770 tau_sun=0.01055 tau_wave6=0.03296 
tau_smk7rate=-0.1792  
 
alpha0=1.43 alpha_tue=0 alpha_wed=0 alpha_thu=0 alpha_fri=0 alpha_sat=0 
alpha_sun=0 alpha_wave6=0 alpha_Smk7rate=0 
 
corr=0 ln_varScale=0; 
 

In addition to the initial values, we also need to specify the prior distribution for the 
parameters. For the beta parameters, which are mutually independent and assumed to have 
the same univariate prior distribution, we can simply specify: 

 
PRIOR be: ~NORMAL (0, VAR=10000); 
 

For a single parameter: 

 
PRIOR cov~NORMAL (0, VAR=10000); 
 

PROC MCMC has flexible options for different specification of the priors. The priors we 
assigned to the parameters in this example are extremely flat priors, a normal distribution 

with extremely large variance, which is close to the (improper) uniform prior which contains 
almost no prior information for the parameters. As a result of this, estimation will not be 
seriously influenced by the prior.  

The mean, BS variance, WS variance are actually specified the same as in PROC NLMIXED. 
However, PROC MCMC doesn’t allow the random effects to be parameter-dependent. So 
before the specification of the mean/BS variance/WS variance, one extra step using the 
Cholesky decomposition is needed: 

 
L11=SQRT(varBS);  
L21=cov/L11; 
L22=SQRT(varScale-L21**2); 

 
By doing the Cholesky decomposition, we represent the location and scale random effects 
as two independent subject-level random effects (𝑠 , 𝑠 ) with unit variance, i.e., 

 

𝜎 𝜎 /𝜎 

𝜎 /𝜎  𝜎 − 𝜎 /𝜎  

𝑠
𝑠 ~ 𝑁

0

0
,

𝜎 𝜎

𝜎  𝜎 

 

 

And, 

𝑠
𝑠 ~ 𝑁

0

0
,

1 0
0  1

 

 

Therefore,  

nu=    L11*sub_1; 
omega= L21*sub_1+L22*sub_2; 
 
mu  = beta0 + beta_tue*tue  + beta_wed*wed  + beta_thu*thu  + beta_fri*fri 
+ beta_sat*sat  + beta_sun*sun  + beta_wave6*wave6  + 
beta_Smk7rate*Smk7rate + nu; 
 
varWS = EXP(tau0 + tau_tue*tue + tau_wed*wed + tau_thu*thu + tau_fri*fri + 
tau_sat*sat + tau_sun*sun + tau_wave6*wave6 + tau_Smk7rate*Smk7rate + 
omega); 
 

Here, the random effects follow a multivariate (bivariate) normal distribution in which the 
mean vector and the covariance matrix are required to be arrays as follows. If the array 
contains all numeric values, parentheses and commas will be needed:  

 
ARRAY location_scale_mean[2] (0,0); 
ARRAY location_scale_cov[2,2] (1,0,0,1); 
 

If the array consists of parameters or variables in the dataset then no parentheses are 
needed:  

 
ARRAY location_scale_rf[2] sub_1 sub_2; 
 

Then the random effect distribution is specified as: 

RANDOM location_scale_rf~MVN(location_scale_mean, location_scale_cov) 
SUBJECT=id; 

 
Combining these statements provides the PROC MCMC code for estimation of the 
parameters of the MELS model: 

PROC MCMC SEED=9879 NMC=100000 NBI=500 MAXTUNE=50 THIN=25 DIC; 
PARMS beta0=11.4756 beta_tue=-0.6326 beta_wed=-0.7576 beta_thu=-0.6899 
beta_fri=-0.5676 beta_sat=-0.04530 beta_sun=0.4450 beta_wave6=-0.2306 
beta_smk7rate=-1.3321  
   
tau0=2.8726 tau_tue=-0.1332 tau_wed=-0.2225 tau_thu=-0.02743  
tau_fri=0.003630 tau_sat=0.04770 tau_sun=0.01055 tau_wave6=0.03296 
tau_smk7rate=-0.1792  
 
alpha0=1.43 alpha_tue=0 alpha_wed=0 alpha_thu=0 alpha_fri=0 alpha_sat=0 
alpha_sun=0 alpha_wave6=0 alpha_Smk7rate=0 
 
cov=0 ln_varScale=.005; 
 
PRIOR be: ~NORMAL (0, VAR=10000); 
PRIOR al: ~NORMAL (0, VAR=10000); 
PRIOR ta: ~NORMAL (0, VAR=10000); 
PRIOR cov: ~NORMAL (0, VAR=10000); 
PRIOR ln_varScale~NORMAL(0,VAR=10000); 
 
varBS = EXP(alpha0 + alpha_tue*tue + alpha_wed*wed + alpha_thu*thu + 
alpha_fri*fri + alpha_sat*sat + alpha_sun*sun + alpha_wave6*wave6 + 
alpha_Smk7rate*Smk7rate); 
 
varScale = EXP(ln_varScale); 
 

  L11=SQRT(varBS); 
  L21=cov/L11; 
  L22=SQRT(varScale-L21**2); 
 

mu  = beta0 + beta_tue*tue + beta_wed*wed  + beta_thu*thu + beta_fri*fri  + 
beta_sat*sat + beta_sun*sun + beta_wave6*wave6  + beta_Smk7rate*Smk7rate + 
L11*sub_1; 
 
varWS = EXP(tau0 + tau_tue*tue + tau_wed*wed + tau_thu*thu + tau_fri*fri + 
tau_sat*sat + tau_sun*sun + tau_wave6*wave6 + tau_Smk7rate*Smk7rate + 
L21*sub_1+L22*sub_2); 
 

  ARRAY location_scale_mean[2] (0,0); 
ARRAY location_scale_cov[2,2] (1,0,0,1); 

  ARRAY location_scale_rf[2] sub_1 sub_2; 
 
  MODEL TimeHour ~ NORMAL(mu,varWS); 
  RANDOM location_scale_rf~MVN(location_scale_mean,location_scale_cov) 
SUBJECT=id; 
 
RUN; 

Display 4. PROC MCMC: Posterior Means 

 
In PROC MCMC, posterior means serve as the estimates for the parameters. PROC MCMC 
also provides the 95% highest posterior density (HPD) intervals or the 95% credible 
intervals for inference. If the interval doesn’t include 0, then we can conclude that the 
covariate has significant effect on the mean/BS variance/WS variance. Compared to the 
results from PROC NLMIXED, PROC MCMC produced similar estimates as well as similar 
conclusions to PROC NLMIXED.  

In the trace plots below, we only show the convergence of the intercepts in mean/BS 
variance/WS variance as well as the scale variance and the covariance between location and 
scale. All of the parameters converge.  

Display 5. PROC MCMC: Convergence 

 

CONCLUSION 
One can estimate the parameters of the MELS model using either frequentist (PROC 
NLMIXED) or Bayesian (PROC MCMC) methods.  For both, PROC MIXED can be used to 
provide starting values for many, but not all, of the model parameters.  For this example 
PROC MCMC was more efficient as it only took 8 minutes to complete the 100,000 
iterations, compared to 26 minutes for PROC NLMIXED (QPOINTS=11).  

